

Final Paper

Government Stability and Female Representation: A Bayesian Replication of Krauss & Kröber (2020)*

Pit Rieger
prieger@ethz.ch
19-951-102

June 8, 2021



Abstract

This paper replicates the analysis from a recent paper that establishes the link between the share of women in cabinets and government stability (Krauss & Kroeber, 2020). Their main finding is that cabinets with higher shares of female ministers tend to stay in office for longer because they are at a lower risk of terminating early. I employ Bayesian exponential and Weibull survival models to accommodate the inherently censored nature of government survival times in democratic systems. My results are very much in line with those of Krauss and Kroeber (2020) and expectations from the literature.

*Replication files for this paper are publicly available on GitHub under github.com/pitrieger/BaysianStats_FinalProject.

1 Introduction & Research Question

Empirical analyses of cabinet stability and duration have a long tradition in political science. Lupia and Strøm (1995) even trace this literature back to early works by Bryce (1921) and Lowell (1896). Much more recently, a paper by Krauss and Kroeber (2020) addresses the research question *whether cabinets with higher shares of women are more stable*. Their findings suggest that indeed the share of women is negatively associated with the risk of early termination. In other words, they find that cabinets with higher shares of female ministers tend to last longer. In this paper, I seek to address the same research question by replicating Krauss and Kroeber's (2020) findings from a Cox proportional hazards regression in a Bayesian framework. More specifically, I rely on exponential and Weibull survival models to do so. Generally, my results are strongly in line with the findings by Krauss and Kroeber (2020).

To give a brief overview of the literature, I briefly summarize the key factors that are associated with government stability and early terminations. In this regard, Schleiter and Morgan-Jones (2009) differentiate between government-specific, parliament-specific, and political-system-specific attributes. Government-specific attributes include the minority status of the government and whether it is a single-party or coalition government (Strøm & Swindle, 2002). Parliament-specific attributes refer to the fragmentation and polarization of the party system (King et al., 1990). Finally, the political system matters for example by determining the duration of the constitutional inter-election period (Strøm & Swindle, 2002). Moreover, with regard to constitutional power, Schleiter and Morgan-Jones (2009) show that systems in which cabinets can dissolve themselves are at a higher risk of early termination (see also Strøm & Swindle, 2002). Several measures of these components are used as control variables by Krauss and Kroeber (2020) and are discussed in greater detail along with the direction of their hypothesized effect in the Data section below.

With respect to the theoretical foundation of Krauss and Kroeber's (2020) research question, they start by arguing that conflict lies at the heart of most factors that are associated with government stability. They further stipulate that different leadership styles determine the response to situations of conflict (Krauss & Kroeber, 2020). In this regard, a broad literature has found that on average, women are more likely to compromise and collaborate to dissolve conflict than men (e.g. Kellerman & Rhode, 2017). Taken together, these arguments lead them to "hypothesise that the presence of a female prime minister and female ministers positively impacts cabinet stability" (Krauss & Kroeber, 2020, p. 4).

I begin the remainder of this paper by describing the data along with several descriptive statistics. I then introduce the exponential and Weibull survival model and discuss the priors for both models. Before turning to the presentation and interpretation of the results, I discuss the convergence diagnostics. Finally, I conclude with a brief summary as well as a short discussion of further extensions and alternative approaches to the models used in this paper.


2 Methods & Data

2.1 Data

I will rely entirely on the replication data from Krauss and Kroeber (2020), which is available online and mostly stems from the ERDDA (Andersson et al., 2014). It includes data on 676 governments in 27 European countries¹. Since I rely on this replication data, this section merely describes their specification of the data.

The main variable of interest $Y_i, i = 1, \dots, n$, is a right-censored random variable, representing the number of days that a given government lasted in office. Of the 676 total observations, 284 were terminated early and the remaining 392 are censored observations. Censoring takes place because for some governments, we cannot observe when they would have collapsed because they survived until the next regular election date. However, we implicitly assume that all governments would fail eventually as time $t \rightarrow \infty$, such that $Y_i < \infty$. In the following, F_i is an indicator of whether government i failed before the next regulation and consequently $C_i = 1 - F_i$ an indicator of whether observation i is censored. Figure 1 provides an overview of the distribution of Y . Clearly, the duration is much longer among the censored cases compared to the cases where an early termination (failure) was observed. In fact, the median duration for the respective groups is 740 and 430.

Figure 1: Distribution of Y split by censorship status of observations



With respect to the substantive research question, Krauss and Kroeber's (2020) derive several covariates from the literature. I mostly follow their specification of the regression

¹Full list of countries: Austria, Belgium, Denmark, Finland, France, Germany, Greece, Iceland, Ireland, Italy, Luxembourg, Netherlands, Norway, Portugal, Spain, Sweden, Great Britain, Bulgaria, Czechia, Estonia, Hungary, Latvia, Lithuania, Poland, Romania, Slovakia, Slovenia.

equation with some differences, discussed below. Table 1 provides summary statistics of the covariates. The main independent variable is the share of female ministers and the gender of the head of government. Due to the fact that Krauss and Kroeber (2020) find that the latter doesn't matter much and that very few cabinets have a female head of government to begin with, I restrict my analysis to the share of female ministers in the cabinet (Model 2 in their paper). This variable is computed across all ministers in a given government, irrespective of the duration of their tenure in government. The finding and expectation of Krauss and Kroeber (2020) is that a higher share is associated with higher government stability. Further covariates are the ideological divisiveness, the effective number of parliamentary parties (ENPP), the length of the constitutional inter-election period (CIEP), the minority status of the cabinet and the maximum possible duration of the cabinet at the time of its formation. The ideological divisiveness is calculated by the maximum pairwise absolute distance between the left-right positions of all coalition parties where the left-right positions were taken from the Comparative Manifesto Project (Volkens et al., 2018). In this regard, the expectation is that ideologically divided coalitions are less stable than single-party governments or coalitions between ideologically close parties. The ENPP is measured as defined by Laakso and Taagepera (1979) and was taken from ParlGov (Döring & Manow, 2020). Again, Krauss and Kroeber (2020) expect and find that the ENPP is associated with low government stability, because there are more options for alternative coalitions as well as more complex coalition bargaining (c.f. Saalfeld, 2008). The CIEP is a measure of the constitutional inter-election period in years and stems from Müller and Strøm (2008), Krauss (2018), as well as Krauss and Kroeber's (2020) own coding. The minority status of a cabinet is a simple indicator variable and was taken from ParlGov (Döring & Manow, 2020). Krauss and Kroeber (2020) find that minority governments are generally at much greater risk of a short duration. However, the authors fail to mention that this is likely the result of the fact that minority government are often regarded as a last-resort option, e.g. when other alternatives have collapsed. A different literature argues that minority governments are just as stable, if not more stable than other types of government [CITE Oli, etc.]. Finally, the maximum possible duration of the cabinet at the time of its formation is measured in days to control for governments that come into office when there isn't much time left before the next scheduled election. The underlying data stems from the ERDDA (Andersson et al., 2014) and ParlGov (Döring & Manow, 2020). Obviously, these cabinets will be terminated sooner (Krauss & Kroeber, 2020).

In some cases, I deviate from Krauss and Kroeber's (2020) operationalization and specification. First, I additionally control for the decade in which the government was inaugurated. The reason for this is that recent governments clearly have higher shares of female ministers, so the decade has the potential to be a confounder. Second, I treat the CIEP as a categorical instead of a numerical variable. The variable essentially reduces to cases with 3, 4, or 5 year cases, rendering its treatment as a continuous variable questionable. Third, I normalize the maximum duration for computational reasons. Finally, I do not adopt the use of time-varying coefficients in the model. Further extensions of the models described

Table 1: Summary of covariates

	Percentage	Mean	SD	Range
Share of Women in Cabinet		13.389	13.427	0 – 55
Ideological Divisiveness		18.42	20.07	0 – 121.87
ENPP		4.04	1.42	2 - 10.8
Max. Poss. Duration (normalized)		0	1.000	-2.595 – 1.627
Decade: 1940	4.6			
Decade: 1950	11.2			
Decade: 1960	9.8			
Decade: 1970	11.7			
Decade: 1980	12.6			
Decade: 1990	18.8			
Decade: 2000	18.5			
Decade: 2010	12.9			
CIEP: 3	0.7			
CIEP: 4	74.6			
CIEP: 5	24.7			
Minority	33.1			

below should look into implementing this feature for survival models in a Bayesian framework.

2.2 Models

2.2.1 General Survival Model Setup

Survival models are still often associated with a frequentist approach. Certainly versatile and easy-to-use implementation of frequentist survival models in R-packages, such as `Survival` (Therneau, 2020) have played a role in their popularity. At the same time, `rstanarm` (Goodrich et al., 2020) has only recently extended its range to survival models in the development version². Instead of relying on this à-la-carte option, I wrote the Stan code myself. However, Eren M. Elçi’s blog post³ on Bayesian survival models was incredibly helpful to vectorize the posterior draws using Stan’s convenient logarithmic complementary cumulative distribution function (`*_lccdf`).

If it wasn’t for the censoring, discussed in the Data section above, survival models could simply be estimated with standard regression models, such as gamma or Weibull regression. It is easy to see that simply ignoring censoring yields biased estimates. To illustrate, suppose we have right-censored data and ignore it in our analysis. In this example, we therefore treat the censor times as actually observed failure times. However, they are in fact merely the lower bounds of failure times for the observations. To deal with right-censoring appropriately, we can instead integrate a censored observation on the interval

²<https://github.com/stan-dev/rstanarm/tree/feature/survival> (retrieved on 11 May 2021)

³https://ermeel86.github.io/case_studies/surv_stan_example.html (retrieved on 11 May 2021)

between its lower bound y_i and infinity. Thus, the likelihood for a censored observation j is given by

$$\begin{aligned}\mathcal{L}_j(\theta) &= \int_{y_j}^{\infty} p(u_j|\theta_j) du_j \\ &= 1 - \mathbb{P}_{\theta}(Y_j \leq y_j),\end{aligned}$$

which is the complementary cumulative distribution function (ccdf) of Y_j .

Thus, the likelihood of all observations can be written as

$$\mathcal{L}_{1:n}(\theta) = \prod_{i=1}^n \{C_j(1 - \mathbb{P}_{\theta_i}(Y_i \leq y_i)) + (1 - C_i)p(y_i|\theta)\}$$

or more conveniently after arranging the data into censored and uncensored observations of length n_1 and $n_2 = n - n_1$

$$\begin{aligned}\mathcal{L}_{1:n}(\theta) &= \mathcal{L}_{1:n_1}(\theta) + \mathcal{L}_{(n_1+1):n}(\theta) \\ &= \prod_{i=1}^{n_1} (1 - \mathbb{P}_{\theta_i}(Y_i \leq y_i)) \prod_{i=n_1+1}^n p(y_i|\theta_i).\end{aligned}\tag{1}$$

2.2.2 Exponential Survival Model

Building on the general survival setup, discussed in the previous section, we require a distributional assumption for the survival times of Y . A first approach is to assume that these survival times are exponentially distributed with probability density function (pdf)

$$p(y_i|\theta_i) = \theta_i \exp(-\theta_i y_i)$$

and cumulative distribution function (cdf)

$$\mathbb{P}_{\theta_i}(Y_i \leq y_i) = 1 - \exp(-\theta_i y_i)$$

such that the ccdf is simply

$$1 - \mathbb{P}_{\theta_i}(Y_i \leq y_i) = \exp(-\theta_i y_i).$$

In all cases, the covariates are included in the rate (inverse scale) parameter $\theta_i \in \mathbb{R}^+$ which is defined via the log link function as

$$\boldsymbol{\theta} = \exp(\beta_0 + \mathbf{X}\boldsymbol{\beta}),$$

where $\exp(\cdot)$ ensures that θ_i is positive.

2.2.3 Weibull Survival Model

Another alternative is to change the distributional assumption to a Weibull distribution. This relaxes the assumption of a constant hazard rate that is made by the exponential model discussed above. Instead, the Weibull model allows for hazards to increase or decrease over time.

No changes are necessary with respect to the general approach of integrating over censored observations, so the likelihood in equation 1 remains. However, θ_i needs to be redefined to accommodate the two parameters of the Weibull distribution. I thus redefine $\theta_i = (\alpha, \sigma_i)$, $\alpha, \sigma \in \mathbb{R}^+$. Further, the pdf of the Weibull distribution is given by

$$p(y_i|\theta_i) = \frac{\alpha}{\sigma_i} \left(\frac{y_i}{\sigma_i} \right) \exp \left(- \left(\frac{y_i}{\sigma_i} \right)^\alpha \right)$$

its cdf by

$$\mathbb{P}_{\theta_i}(Y_i \leq y_i) = 1 - \exp \left(- \left(\frac{y_i}{\sigma_i} \right)^\alpha \right)$$

and so its ccdf is

$$1 - \mathbb{P}_{\theta_i}(Y_i \leq y_i) = \exp \left(- \left(\frac{y_i}{\sigma_i} \right)^\alpha \right).$$

I then follow Bueros et al.'s (2021) inclusion of covariates by parametrizing

$$\boldsymbol{\sigma} = \exp \left(- \frac{\beta_0 + \mathbf{X}\boldsymbol{\beta}}{\alpha} \right).$$

2.3 Prior Selection

Both the exponential and the Weibull model require priors for the intercept β_0 and the coefficients of the covariates $\boldsymbol{\beta}$. Since in both cases, these are simple regression coefficients, I use jointly independent normal priors for all parameters. Formally,

$$\beta_0 \sim \mathcal{N}(0, 100^2)$$

$$\boldsymbol{\beta} \sim \mathcal{N}_p(0, 5^2 I),$$

where \mathcal{N}_p is the multivariate normal distribution and I a $p \times p$ identity matrix. Note that the prior is much less informative for the intercept compared to the covariates. The normal distribution was chosen because it is the standard choice for regression coefficients. Furthermore, I chose a more uninformative prior for the intercept as to not restrict the baseline hazard too much. However, recall that the models are using a log link. As a result, the priors on the covariate coefficients are still relatively uninformative.

Finally, the Weibull model requires an additional prior for the shape parameter α . There seems to be no clear common practice with respect to the prior selection in the literature. However, we require $\alpha \in \mathbb{R}^+$. I therefore place a very wide half-Cauchy prior with location 0 and scale 100 on α , i.e.

$$\alpha \sim \text{Cauchy}^+(0, 100).$$

However, cursory checks (not included here) with much narrower priors show that the posterior of α is not very sensitive.

3 Analysis

3.1 Convergence

Both the exponential and the Weibull model are fitted with 4000 iterations on 4 Markov chains. The first 2000 iterations of every chain are used as a warm-up and are not used for the posterior estimates. Figure 2 shows the trace plots of all four chains for the regression parameters in the exponential survival model. Analogously, Figure 3 traces the four chains for the Weibull model. Note that the latter also includes the α parameter on top of the regression parameters. Clearly, both figures show that the chains have mixed well, which is also indicated by the \hat{R} -statistic, which is 1.0 for all parameters in both models (not included here). This suggests that the models have converged and that we can trust the posterior estimates from the models for the remainder of this paper.

Figure 2: Traces of the parameters in the Exponential survival model.






Figure 3: Traces of the parameters in the Weibull survival model.



3.2 Results


Beginning with the posteriors from the exponential model, Figure 4 shows that several covariates are clearly associated with a higher or lower government stability. Generally, negative effects are indicative of a higher government stability and positive effects of earlier termination. This can be verified by considering the expectation of our dependent exponential random variable $Y_i \sim \text{Expo}(\theta_i)$ which is given by $\mathbb{E}[Y_i] = 1/\theta_i$. Decreases in the linear predictor $\beta_0 + \mathbf{X}_i\beta$ thus lead to decreases in θ_i and finally an increase in $\mathbb{E}[Y_i]$.

Figure 4: Posterior densities of the coefficients in the exponential survival model.



The posteriors from the exponential model are in line with Krauss and Kroeber’s (2020) main finding that cabinets with higher shares of women also tend to last longer. The magnitude of this effect is further explored below. Similarly, the longer the maximum possible duration at the time of government formation, the higher the stability of the government. In the opposite direction, minority governments, ideologically divided cabinets and cabinets arising from legislatures with a high effective number of parliamentary parties (ENPP) are clearly more likely to terminate sooner. With respect to the constitutional inter-election period (CIEP), the interpretation is more ambiguous. While longer CIEPs tend to be associated with lower stability, both densities cover the null comfortably. However, recall from Table 1 that the reference group with a CIEP of 3 years only consists of a very small number of cabinets. Finally, the results also show that recent cabinets, particularly in the 2000s and 2010s, tend to be less stable than cabinets in earlier decades.


Figure 5: Posterior predictive distribution for different shares of women in cabinet in the exponential survival model.



To further explore the main focus of Krauss and Kroeber's (2020) and this paper, Figure 5 and 6 visualize the magnitude of the effect of the share of women in cabinet. Figure 5 depicts the posterior predictive distributions for different shares of women in the cabinet. The minimum, mean, and maximum reflect the sample values of female cabinet ministers and correspond to 0%, 13.4%, and 54.5%, respectively. The remaining variables are set to encode a non-minority cabinet in the 2010's in a system with a CIEP of 4 years while holding all numeric variables constant at their respective sample mean. Clearly, the expected government duration increases substantively as a function of the share of women in cabinet. To illustrate, the median duration for the minimum, mean, and maximum share of women in cabinet is about 605, 985, and 4360 days, respectively. This can also be gathered from the survival functions for the same groups of cabinets, shown in Figure 6. For cabinets with high shares of women in the setting described above, the survival function is barely decreasing, indicating very high stability. For cabinets without any women on the other hand, the survival function decreases with less than 12.5% of cabinets expected to remain in office at 2000 days, the empirical maximum duration (of course including right-censoring).

However, the shape of the survival functions in Figure 6 also reveals a weakness of assuming exponential survival times: The functional form of the exponential survival function, an implication of a constant hazard of termination, is theoretically questionable. Especially the steep initial decrease in the survival functions seems unrealistic. It seems more likely that cabinets what have a high probability of terminating very early are less likely to be formed in the first place. A survival function that reflects this would need to decrease slowly at first and then begin to decrease more at a later stage. Therefore, it is prudent to have a look at the more flexible Weibull distribution, which can accommodate such

Figure 6: Survival function of the exponential survival model.



Note: Shaded areas represent 95% credible intervals, taking into account only the posterior variance of the coefficient for the share of women in cabinet. As a result, the purple line for the minimal share of women in cabinet doesn't have a shaded area, because that share is equal to zero. Instead, it be interpreted as a baseline survival function.

survival function shapes. Convergence diagnostics and a coefficient plot, analogous to Figures 2 and 4, can be found in the appendix, but were excluded from the main section because the results do not differ substantially from Figures 2 and 4. In fact, with respect to the coefficients of the covariates, the models are strikingly similar and no noteworthy inferential differences exist.

Finally, Figures 7 and 8 are analogous to Figures 5 and 6, but were created from the Weibull model instead. They both indicate that there's some merit to the argument made above. The median duration for the minimum, mean, and maximum share of women in cabinet according to the posterior predictive distribution of the Weibull model, as shown in Figure 7, is about 610, 915, and 3340 days, respectively. Furthermore, the survival functions in Figure 8 indeed don't decrease as steeply as in the exponential model. However, generally speaking, the difference between the Weibull model and the exponential model are rather minor.

Figure 7: Posterior predictive distribution for different shares of women in cabinet in the Weibull survival model.





Figure 8: Survival function of the Weibull survival model.



Note: Shaded areas represent 95% credible intervals, taking into account only the posterior variance of the coefficient for the share of women in cabinet. As a result, the purple line for the minimal share of women in cabinet doesn't have a shaded area, because that share is equal to zero. Instead, it be interpreted as a baseline survival function.

3.3 Conclusion

In this paper, I have replicated the analysis from Krauss and Kroeber (2020) in a Bayesian framework with exponential and Weibull survival models. My results corroborate their main finding that cabinets with higher shares of female ministers are indeed more likely to stay in office longer by being less prone to early termination. I further show that there is not much of a difference between assuming exponentially or Weibull distributed survival times. Particularly with respect to inference, the coefficients of the covariates do not change substantially. Moreover, findings with respect to a selection of control variables are mostly in line with the literature.

Further work could explore the use of time-varying coefficients for the ideological divisiveness and the CIEP that Krauss and Kroeber (2020) use. So far these features have not been implemented in my models. However, I find that when controlling for the decade, something that Krauss and Kroeber (2020) only do as a robustness test, the assumption of proportional hazards for a frequentist Cox proportional hazard model is not rejected by standard tests.

While writing this paper, I also explored the possibility of modelling a transformed version of the outcome as a zero-inflated Poisson model. Details are included in the appendix and code is available in the GitHub repository of this paper. However, there were several convergence issues when modelling the zero-inflation component with the full set of covariates $\mathbf{Z} = \mathbf{X}$. These issues do not arise for certain subsets $\mathbf{Z}' \subset \mathbf{Z}$ for the zero inflation and the full set of covariates \mathbf{X} for the Poisson component. In this case, the model yields results that are equivalent to the survival models discussed in this paper. Further work is necessary to fix the issue of non-convergence, for instance by finding a better parametrization of the model.

References

- Andersson, S., Bergman, T., & Ersson, S. (2014). The european representative democracy data archive, release 3 [Main sponsor: Riksbankens Jubileumsfond]. www.erdda.se
- Bryce, J. (1921). *Modern democracies*. Macmillan.
- Buros, J., Novik, E., & Goodrich, B. (2021). *Biostan: Intro to stan for bioc2016 workshop* [R package version 0.1.0]. <http://github.com/jburos/biostan/>
- Döring, H., & Manow, P. (2020). Parliaments and governments database (ParlGov): Information on parties, elections and cabinets in modern democracies. development version.
- Goodrich, B., Gabry, J., Ali, I., & Brilleman, S. (2020). Rstanarm: Bayesian applied regression modeling via stan [version 2.21.2]. <https://mc-stan.org/rstanarm>
- Kellerman, B., & Rhode, D. L. (2017). *Women and leadership: The state of play and strategies for change*. Oxford University Press.
- King, G., Alt, J. E., Burns, N. E., & Laver, M. (1990). A unified model of cabinet dissolution in parliamentary democracies. *American Journal of Political Science*, 34(3), 846–871. <http://www.jstor.org/stable/2111401>
- Krauss, S. (2018). Stability through control? the influence of coalition agreements on the stability of coalition cabinets. *West European Politics*, 41(6), 1282–1304. <https://doi.org/10.1080/01402382.2018.1453596>
- Krauss, S., & Kroeber, C. (2020). How women in the executive influence government stability. *Journal of European Public Policy*, 0(0), 1–19. <https://doi.org/10.1080/13501763.2020.1773905>
- Laakso, M., & Taagepera, R. (1979). "effective" number of parties: A measure with application to west europe. *Comparative Political Studies*, 12(1), 3–27. <https://doi.org/10.1177/001041407901200101>
- Lowell, A. L. (1896). *Governments and parties in continental europe*. Harvard University Press.
- Lupia, A., & Strøm, K. (1995). Coalition termination and the strategic timing of parliamentary elections. *American Political Science Review*, 89(3), 648–665. <https://doi.org/10.2307/2082980>
- Müller, W. C., & Strøm, K. (2008). Coalition agreements and cabinet governance. In K. Strøm, W. C. Müller, & T. Bergman (Eds.), *Cabinets and coalition bargaining: The democratic life cycle in western europe* (pp. 159–199). Oxford University Press.
- Saalfeld, T. (2008). Institutions, chance and choices: The dynamics of cabinet survival. In K. Strøm, W. C. Müller, & T. Bergman (Eds.), *Cabinets and coalition bargaining: The democratic life cycle in western europe* (pp. 327–368). Oxford University Press.
- Schleiter, P., & Morgan-Jones, E. (2009). Constitutional power and competing risks: Monarchs, presidents, prime ministers, and the termination of east and west european cabinets. *The American Political Science Review*, 103(3), 496–512. <http://www.jstor.org/stable/27798518>
- Strøm, K., & Swindle, S. M. (2002). Strategic parliamentary dissolution. *The American Political Science Review*, 96(3), 575–591. <http://www.jstor.org/stable/3117931>

- Therneau, T. (2020). Survival: A package for survival analysis in r [version 3.2-7]. <https://CRAN.R-project.org/package=survival>
- Volkens, A., Krause, W., Lehmann, P., Matthieß, T., Merz, N., Regel, S., & Weßels, B. (2018). The Manifesto Data Collection. Manifesto Project (MRG/CMP/MARPOR) [Version 2018b.]. <https://doi.org/10.25522/manifesto.mpd.2018b>

4 Appendix

4.1 Stan Code

This section contains the Stan model code for both the exponential and Weibull survival model. Additional R-code is available on GitHub.

Listing 1: Exponential Survival Model

```
1 data {
2   int<lower=0> n_event; // # uncensored obs
3   int<lower=0> n_censor; // # censored obs
4   int<lower=0> p; // # covariates excluding intercept
5   matrix[n_event,p] X_event; // uncensored obs design matrix
6   matrix[n_censor,p] X_censor; // censored -----"-----
7   matrix[3,p] X_pred; // for posterior predictive distribution
8   vector<lower=0>[n_event] T_event; // uncensored survival times
9   vector<lower=0>[n_censor] T_censor; // lower bound of censored survival
    times = censor time
10 }
11 parameters {
12   real alpha;
13   vector[p] beta;
14 }
15 transformed parameters{
16   vector[n_event] lambda_event = exp(alpha + X_event * beta);
17   vector[n_censor] lambda_censor = exp(alpha + X_censor * beta);
18 }
19 model {
20   alpha ~ normal(0, 100);
21   beta ~ normal(0, 5);
22   target += exponential_lpdf(T_event | lambda_event);
23   target += exponential_lccdf(T_censor | lambda_censor);
24 }
25 generated quantities {
26   vector[n_event + n_censor] T_hat;
27   vector[3] T_pred;
28   for(i in 1:n_event){
29     T_hat[i] = exponential_rng(lambda_event[i]);
30   }
31   for(i in 1:n_censor){
32     T_hat[i + n_event] = exponential_rng(lambda_censor[i]);
33   }
34   for(i in 1:3){
35     T_pred[i] = exponential_rng(exp(alpha + X_pred * beta)[i]);
36   }
37
38 }
```


Listing 2: Weibull Survival Model

```
1 data {
2   int<lower=0> n_event; // # uncensored obs
3   int<lower=0> n_censor; // # censored obs
4   int<lower=0> p; // # covariates excluding intercept
5   matrix[n_event,p] X_event; // uncensored obs design matrix
6   matrix[n_censor,p] X_censor; // censored -----"-----
7   matrix[3,p] X_pred; // for posterior predictive distribution
8   vector<lower=0>[n_event] T_event; // uncensored survival times
9   vector<lower=0>[n_censor] T_censor; // lower bound of censored survival
    times = censor time
10 }
11 parameters {
12   real mu;
13   vector[p] beta;
14   real<lower = 0> alpha;
15 }
16 transformed parameters {
17   vector[n_event] sigma_event = exp(- (mu + X_event * beta)/ alpha);
18   vector[n_censor] sigma_censor = exp(- (mu + X_censor * beta)/ alpha);
19 }
20
21 model {
22   mu ~ normal(0, 100);
23   beta ~ normal(0, 5);
24   alpha ~ cauchy(0, 2.5);
25   target += weibull_lpdf(T_event | alpha, sigma_event);
26   target += weibull_lccdf(T_censor | alpha, sigma_censor);
27 }
28 generated quantities {
29   vector[n_event + n_censor] T_hat;
30   vector[3] T_pred;
31   for(i in 1:n_event){
32     T_hat[i] = weibull_rng(alpha, sigma_event[i]);
33   }
34   for(i in 1:n_censor){
35     T_hat[i + n_event] = weibull_rng(alpha, sigma_censor[i]);
36   }
37   for(i in 1:3){
38     T_pred[i] = weibull_rng(alpha, exp(- (mu + X_pred * beta)/ alpha)[i]);
39   }
40 }
```

4.2 Weibull Survival Model

This section contains the coefficient plot for the Weibull model, which was excluded from the main section because it does not differ substantially from the coefficient plot for the exponential model, shown in Figure 4.

Figure A1: Posterior densities of the coefficients in the Weibull survival model.



4.3 Zero-Inflated Poisson Regression

Since the main interest of Krauss and Kroeber (2020) and this paper does not lie in the concrete survival times, but rather on the association between government stability and the share of female ministers in the cabinet, the problem can also be approached differently, i.e. outside the survival modelling framework. It is generally known when the next regular elections will be held. At the time of government formation, we can thus already tell what the maximum number of possible days in office for that government will be. In fact, this is used as one of the covariates in the survival models in the paper. Denoting this maximum as m_i , we can compute the number of days that the government fell short of reaching that maximum as

$$\tilde{Y}_i = m_i - Y_i.$$

We thus have a count variable \tilde{Y}_i that can be modelled as a zero-inflated Poisson random variable. The likelihood of this model is therefore given by

$$\mathcal{L}_i(\theta) = \begin{cases} \pi_i + (1 - \pi_i)e^{-\lambda_i}\lambda_i^0 & , \tilde{y}_i = 0 \\ (1 - \pi_i)\frac{e^{-\lambda_i}\lambda_i^{\tilde{y}_i}}{\tilde{y}_i!} & , \text{otherwise} \end{cases}$$

where $\theta = (\pi, \lambda)$ can be defined via logistic and log link functions, respectively, such that

$$\begin{aligned}\pi &= (1 + \exp(-(\beta_0 + \mathbf{X}\boldsymbol{\beta})))^{-1} \\ \lambda &= \exp(\gamma_0 + \mathbf{Z}\boldsymbol{\gamma}).\end{aligned}$$

`Stan` and `R`-code for this model are included in the GitHub repository of this paper. However, the parameters of the zero-inflation component fail to converge properly when using the full set of covariates for both \mathbf{X} and \mathbf{Z} . When using certain subsets $\mathbf{Z}' \subset \mathbf{Z}$ and the full set of covariates \mathbf{X} , the model yields results that are equivalent to the survival models discussed in this paper.